


0 SciLab'i testülesanne

1. Joonestame järgmise siilikese.



2. Põhijoon on kirjeldatav polaarkoordinaatides valemiga

$$r(t) = \cos(44t) + 2 \sin(t), \quad t \in [0, \pi]. \quad (0.1)$$

Selleks, et antud joont saaks kuvada tavalisel xy -tasandil (x väärtused vastu y väärtusi), tuleks teha koordinaatide teisendus

$$\begin{cases} x = r(t) \cos(t) \\ y = r(t) \sin(t) \end{cases}. \quad (0.2)$$

3. Lahenduse idee ise on järgmine. Moodustame näiteks t väärtuste vektori

$$t = [0, 1, 2, 3] \Rightarrow r(t) = [r(0), r(1), r(2), r(3)],$$

millest

$$\cos(t) = [\cos(0), \cos(1), \cos(2), \cos(3)].$$

4. Polaarkoordinaatidelt üleminek ristkoordinaatidele.

Nüüd peab jälgima, et SciLab'is tehakse vaikinisi kõik tehted maatriksitega (vektoritega). Näiteks ei saa me korrutada omavahel $t * t$, kuna maatriksite korrutamise reegel seda ei luba (dimensioonid on valed). Korrutades $t * t'$ saame 1×1 maatriksi (1×4 ja 4×1), korrutades $t' * t$ saame 4×4 maatriksi (4×1 ja 1×4).

Suureks abiks on punktiviisiline korrutamine “. * “. Siis

$$\begin{aligned} r(t) .* \cos(t) &= [r(0), r(1), r(2), r(3)] .* [\cos(0), \cos(1), \cos(2), \cos(3)] \\ &= [r(0)\cos(0), r(1)\cos(1), r(2)\cos(2), r(3)\cos(3)]. \end{aligned}$$

5. Näo osa juures kasutame lihtsalt punktide graafikule kandmist.

0.1 SciLab, lahendus

```
clear // Lihtsalt kasulik käsk, tühjendab mälu
// Sissejuhatav ülesanne. Marek Kolk, 2013–15.
// Joonistame siili
// Joon on antud polaarkoordinaatides, teisendame ta xy-tasandile

function [ xx ] = xtelg( t , raadius ) // x-telje väärtused
    xx = raadius( t ) .* cos( t ) // punktiviisiline *
endfunction

function [ yy ] = ytelg( t , raadius ) // y-telje väärtused
    yy = raadius( t ) .* sin( t )
endfunction

function [ yy ] = raadius( t ) // Joon polaarkoordinaatides
    yy = cos( 44*t )+2*sin( t )
endfunction

// Loomenurga t väärtused reavektorina, t muutub nullist piini
t = [ 0 : %pi/720 : %pi ]
x = xtelg( t , raadius ) // x-telje väärtused
y = ytelg( t , raadius ) // y-telje väärtused
// Graafikuga seonduv, saab ka lühemalt ja saab lihtsamini
gcf() // Graafiku järjehoidja
scf(0) // Tee aktiivseks aken nr 0
clf(0) // Puhasta aken nr 0

plot( x , y , "r-" )
plot( x , y , "g*" )
plot( 0.1 , 0.6 , "bpentagram" )
plot( -0.1 , 0.6 , "bpentagram" )
plot( 0 , 0.4 , "k." )
suu = [-0.1 : 0.01 : 0.1 ]
plot( suu , (2*suu).^2 +0.25 , "k-" )
plot( 0 , 3.5 , "wo" )
```